

I2S DAC


A PCM5102A that turns I2S samples from an ESP32 into stereo line level on a 3.5 mm jack.

✗ Line level: never a bare speaker, never 5 V

The jack is for **headphones, powered speakers or an amplifier's input**. A bare speaker needs the TK100. The chip runs from 3.0 to 3.6 V: power it from 3V3, never 5 V, which is over its 3.9 V maximum.

Pins, and which to wire

Drawn from above, jack at the top. BCK, DIN and LRCK come from the ESP32; the chip builds its own system clock from BCK. XSMT is pulled HIGH on the board, so it plays with that pin unconnected.



The bit clock must be 32 or 64 per sample

With no system clock, the chip locks its PLL to BCK. At 44.1 and 48 kHz, 16-bit stereo gives 32 per sample and locks. At 16 kHz TI lists only 64, so send 32-bit slots. At 8 kHz nothing locks.

Sample rate and slot, and whether it locks

44.1 kHz, 16-bit

1.4112 MHz on BCK, 32 per sample

On TI's list: **locks**. Music and most MP3s.

16 kHz, 16-bit

0.512 MHz on BCK, 32 per sample

Not on TI's list for 16 kHz. May or may not lock.

16 kHz, 32-bit

1.024 MHz on BCK, 64 per sample

On TI's list: **locks**. Voice from the TK101.

Choose the slot width from the sample rate: 16-bit at 32 kHz and up, 32-bit at 16 kHz. The ESP32 sends BCK at sample rate $\times$ slot bits $\times$ 2.

Table 11 of TI's PCM510xA datasheet (SLAS859C), for the PLL with no system clock, which is how this board runs. The frequencies are arithmetic. Sketch compiled for the ESP32-S3 with esp32 core 3.3.10.

Left, then right

Arduino IDE · Board: ESP32S3 Dev Module · USB CDC On Boot: Enabled · No libraries

```
#include <ESP_I2S.h>

const int PIN_BCK = 42, PIN_DIN = 41;
const int PIN_LRCK = 40, PIN_XSMT = 39;
const int RATE = 44100;
I2SClass dac;

void tone2(float hz, int ms, int left, int right) {
  int16_t frame[2];
  float phase = 0, step = 2 * PI * hz / RATE;
  for (long i = 0; i < (long)RATE * ms / 1000; i++) {
    int16_t s = (int16_t)(3000 * sinf(phase));
    frame[0] = left ? s : 0;           // the left sample goes first
    frame[1] = right ? s : 0;
    dac.write((uint8_t *)frame, sizeof(frame));
    phase += step;
    if (phase > 2 * PI) phase -= 2 * PI;
  }
}

void setup() {
  Serial.begin(115200);
  pinMode(PIN_XSMT, OUTPUT);
  digitalWrite(PIN_XSMT, HIGH);      // un-mute
  dac.setPins(PIN_BCK, PIN_LRCK, PIN_DIN);
  dac.begin(I2S_MODE_STD, RATE, I2S_DATA_BIT_WIDTH_16BIT,
            I2S_SLOT_MODE_STEREO);
}

void loop() {
  Serial.println("left");
  tone2(440, 1000, 1, 0);
  Serial.println("right");
  tone2(440, 1000, 0, 1);
  Serial.println("both");
  tone2(440, 1000, 1, 1);
  tone2(0, 1000, 0, 0);
}
```

left
right
both

The Serial Monitor at 115200, each line as the tone moves to that ear.

Headphones, powered speakers

Full scale is 2.1 V RMS, made for inputs of 1 kΩ and up. Headphones work, more quietly: through the board's 470 Ω a 32 Ω earpiece gets about 6 % of the voltage, about 0.13 V RMS at full scale, worked out. For loud, use powered speakers: the kit's 3.5 mm to 3.5 mm cable, or the 3.5 mm to RCA cable into a stereo's line input.

When it does not work

The LED is on and there is no sound

No BCK or LRCK, or XSMT is held LOW. Check the three I2S wires and the XSMT pin.

Left and right are swapped

The left sample goes first in each frame. Swap the two values in the sketch.

A bare speaker is silent

Line level cannot drive it. Use the TK100, or a powered speaker.

Specifications

Chip	PCM5102A (TI), stereo DAC	Supply	3.3 V; 3.0 to 3.6 V
Output	2.1 V RMS full scale, line level	Performance	112 dB dynamic range, typical
Sample rate	8 to 384 kHz; see the bit clock	XSMT	HIGH by 10 kΩ; LOW mutes
Jack	3.5 mm stereo	Board	22.4 × 30.4 mm, header fitted

Wiring

	FRONT	ESP32-S3
1	GND	GND
2	3V3	3V3
3	BCK	GPIO 42
4	DIN	GPIO 41
5	LRCK	GPIO 40
6	XSMT	GPIO 39 (opt.)

XSMT is optional: pulled HIGH on the board, it plays with nothing connected, and a GPIO driving it LOW mutes with a ramp rather than a click. Any free GPIO works for the I2S lines.

Silence at 16 kHz, sound at 44.1

16-bit stereo at 16 kHz is 32 per sample, not on TI's list. Use 32-bit slots.

Very quiet in headphones

Expected: 470 Ω in series. Raise the level in the sketch, or use powered speakers.

Clicks between tunes

Mute with XSMT LOW before stopping the clocks, HIGH after starting them.